



Embedded A.I - Sensor Acquisition and Actuator Control



Installing Libraries

- Open Arduino IDE → Go to:
- Sketch > Include Library > Manage Libraries
- DHT sensor library — by Adafruit
- Adafruit Unified Sensor
- ESP32Servo — by Kevin Harrington, John K. Bennett



DHT11 Sensor

- DHT11 is a basic, low-cost digital sensor
- Measures:
 - Temperature (0–50°C ±2°C)
 - Humidity (20–90% ±5%)
- 3 Pins used:
 - VCC – Power (3.3V or 5V)
 - GND – Ground
 - DATA – Outputs digital signal with temp/humidity

```
#include <DHT.h>
#include <Adafruit_Sensor.h>

#define DHTPIN 5 // Connect DHT11 to GPIO5
#define DHTTYPE DHT11

DHT dht(DHTPIN, DHTTYPE);

void setup() {
  Serial.begin(115200);
  dht.begin();
}

void loop() {
  float temp = dht.readTemperature();
  float hum = dht.readHumidity();

  if (isnan(temp) || isnan(hum)) {
    Serial.println("Failed to read from DHT sensor!");
  } else {
    Serial.print("Temperature: ");
    Serial.print(temp);
    Serial.println(" °C");

    Serial.print("Humidity: ");
    Serial.print(hum);
    Serial.println(" %");
  }

  delay(2000); // Read every 2 seconds
}
```

Servo Motor

- Servo Motor is an actuator that rotates to specific angles
- You can set angles from 0° to 180°
- Internal feedback maintains that position
- 3 Pins used:
 - Orange (Signal) → GPIO (e.g. D6)
 - Red (VCC) → 5V (some work with 3.3V)
 - Brown (GND) → Ground


```
#include <ESP32Servo.h> // Use this library for ESP32
#define SERVOPIN 6 // Connect Servo to GPIO6

Servo myServo;

void setup() {
  Serial.begin(115200);
  myServo.attach(SERVOPIN);
  myServo.write(0); // Initial position
  Serial.println("Servo ready!");
}

void loop() {
  for (int angle = 0; angle <= 180; angle += 10) {
    myServo.write(angle);
    Serial.print("Moving to angle: ");
    Serial.println(angle);
    delay(500);
  }

  for (int angle = 180; angle >= 0; angle -= 10) {
    myServo.write(angle);
    Serial.print("Moving to angle: ");
    Serial.println(angle);
    delay(500);
  }
}
```



Climate-Reactive Gauge

- Read temperature from DHT11
- Rotate servo proportionally:
- Objective
 - 20°C → 0°
 - 30°C → 90°
 - 40°C → 180°

```
#include <Adafruit_Sensor.h>
#include <DHT.h>
#include <ESP32Servo.h>

#define DHTPIN 5
#define DHTTYPE DHT11
#define SERVOPIN 6

DHT dht(DHTPIN, DHTTYPE);
Servo myServo;

void setup() {
  Serial.begin(115200);
  dht.begin();
  myServo.attach(SERVOPIN);
  myServo.write(0); // Reset position
  delay(1000);
}

void loop() {
  float temperature = dht.readTemperature();

  if (isnan(temperature)) {
    Serial.println("Failed to read from DHT11!");
    return;
  }

  Serial.print("Temperature: ");
  Serial.print(temperature);
  Serial.println(" °C");

  // Map temperature (20–40°C) to angle (0–180°)
  int angle = map((int)temperature, 20, 40, 0, 180);
  angle = constrain(angle, 0, 180); // Clamp to range

  myServo.write(angle);
  Serial.print("Servo Angle: ");
  Serial.println(angle);

  delay(2000);
}
```

